

## 1. 平面波の進行波と定在波

**理論** 1次元伝送線路の電圧・電流方程式は、入射振幅を $V_0^+$ 、反射係数を $\Gamma$ 、伝搬定数を $\beta$ 、特性インピーダンスを $Z_0$ とすると次式で与えられる。

$$\begin{cases} V(x) = V_0^+ (e^{j(\omega t - \beta x)} + \Gamma e^{j(\omega t + \beta x)}) \\ I(x) = \frac{V_0^+}{Z_0} (e^{j(\omega t - \beta x)} - \Gamma e^{j(\omega t + \beta x)}) \end{cases} \quad (1)$$

ここで、線路上の電圧振幅の絶対値を考えると次式が得られる。

$$|V(x)| = |V_0^+| |1 + \Gamma e^{2j\beta x}| \quad (2)$$

### スクリプト

\*\*\*\*\* 初期画像を描画するスクリプト例 \*\*\*\*\*

```
# plane_wave.gnu
g=-0.5          # reflection coefficient
pai=acos(-1)    # pai=3.14159...
freq=10.0e9     # frequency [Hz]
cvel=3.0e8      # light velocity [m/s]
w=2.0*pai*freq  # omega [rad/s]
b=w/cvel        # beta [rad/m]
j=sqrt(-1)      # imaginary number
set yrange [-3:3]
set xrange [-0.1:0.1]
set samples 1000
set grid

t=0.0
plot real (exp(j*(w*t-b*x))+g*exp(j*(w*t+b*x))) w l, ¥
abs(1.0+g*exp(2.0*j*b*x)) w l
pause -1 'Hit "Enter key" to start animation...'
load 'animation.gnu'
```

\*\*\*\*\* 初期画像を動かすためのスクリプト \*\*\*\*\*

```
# animation.gnu
t=t+1/freq/50
replot
reread
```

※reread コマンドは直前に load したファイルの内容を再度読み込むコマンドである。

※描画を停止するときは、Ctrl-c を押す。

### 課題

- (1) 反射係数の大きさを変えて（複素数でもよい）実行結果の変化を観察せよ。
- (2) 例題では進行波と反射波の合成振幅を描画している。これを改良して進行波と反射波を分離して表示せよ。
- (3) 合成振幅のみ線の太さを変形せよ。
- (4) 式(1)から式(2)を導出せよ。

## 2. 円筒波の進行波と定在波

**理論** 2次元平面の1点を波源として伝搬する水面波は、入射振幅を1、反射係数を $\Gamma$ 、伝搬定数を $\beta$ とすると次式で与えられる。

$$V(x) = H_0^{(2)}(\beta x)e^{j\omega t} + \Gamma H_0^{(1)}(\beta x)e^{j\omega t} \quad (3)$$

ただし、

$$\begin{cases} H_0^{(1)}(\beta x) = J_0(\beta x) + jY_0(\beta x) \\ H_0^{(2)}(\beta x) = J_0(\beta x) - jY_0(\beta x) \end{cases}$$

なお、波面の位相角度は次式で与えられる。

$$\angle\theta = \tan^{-1}\left(\frac{\text{Im}[V(x)]}{\text{Re}[V(x)]}\right) \quad (4)$$

### スクリプト

\*\*\*\*\* 初期画像を描画するスクリプト例 \*\*\*\*\*

```
# cylindrical_wave.gnu
g=0.0 # reflection coefficient
pai=acos(-1) # pai=3.14159...
freq=10.0e9 # frequency [Hz]
cvel=3.0e8 # light velocity [m/s]
w=2.0*pai*freq # omega [rad/s]
b=w/cvel # beta [rad/m]
j=sqrt(-1) # imaginary number
set yrange [-4:4]
set xrange [0.0:0.2]
set samples 1000
set grid

han0_2(x)=besj0(b*x)-j*besy0(b*x) # zeroth order Hankel function of the second kind
han0_1(x)=besj0(b*x)+j*besy0(b*x) # zeroth order Hankel function of the first kind

fwd(x,t)=han0_2(x)*exp(j*w*t) # traveling wave function
bwd(x,t)=han0_1(x)*exp(j*w*t) # backward wave function

t=0.0
plot real( fwd(x,t)+g*bwd(x,t) ) w l lw 2, ¥
atan2( imag( fwd(x,t)+g*bwd(x,t) ), real( fwd(x,t)+g*bwd(x,t) )) with l 3
pause -1 'Hit "Enter key" to start animation...'
load 'animation.gnu'
```

\*\*\*\*\* 初期画像を動かすためのスクリプト \*\*\*\*\*

```
# animation.gnu
t=t+1/freq/50
replot
reread
```

### 課題

- (1) 反射係数の大きさを変えて（複素数でもよい）実行結果の変化を観察せよ。
- (2) 例題では進行波と反射波の合成波形を描画している。これを改良して進行波と反射波を分離して表示せよ。
- (3) 円筒波の振幅は $1/x$ で減衰するはずである。減衰特性がわかるように $1/x$ の関数も合わせて表示せよ。

### 3. 球面波の進行波と定在波

**理論** 3次元空間の1点を波源として伝搬する球面波は、入射振幅を1、反射係数を $\Gamma$ 、伝搬定数を $\beta$ とすると次式で与えられる。

$$V(x) = h_0^{(2)}(\beta x)e^{j\omega t} + \Gamma h_0^{(1)}(\beta x)e^{j\omega t} \quad (5)$$

ただし、

$$\begin{cases} h_0^{(1)}(\beta x) = j_0(\beta x) + j y_0(\beta x) = \frac{e^{+j\beta x}}{j\beta x} \\ h_0^{(2)}(\beta x) = j_0(\beta x) - j y_0(\beta x) = \frac{e^{-j\beta x}}{-j\beta x} \end{cases}$$

なお、波面の位相角度は次式で与えられる。

$$\angle\theta = \tan^{-1}\left(\frac{\text{Im}[V(x)]}{\text{Re}[V(x)]}\right) \quad (6)$$

#### 課題

- (1) 球面波の動画スクリプトを作成せよ。
- (2) 反射係数の大きさを変えて（複素数でもよい）実行結果の変化を観察せよ。
- (3) 例題では進行波と反射波の合成波形を描画している。これを改良して進行波と反射波を分離して表示せよ。
- (4) 球面波の振幅は $1/x^2$ で減衰するはずである。減衰特性がわかるように $1/x^2$ の関数も合わせて表示せよ。

### 4. 導波管の進行波と定在波

**理論** 導波管を伝搬する $TE_{10}$ 基本モードは、入射振幅を1、反射係数を $\Gamma$ 、伝搬定数を $\beta$ とすると次式で与えられる。

$$E_y(x, y, z) = (e^{-j\beta z} + \Gamma e^{+j\beta z})e_y(x, y)e^{j\omega t} \quad (7)$$

ただし、

$$e_y(x, y) = \sin\left(\frac{\pi x}{a}\right)$$

#### 課題

- (1) 導波管 $TE_{10}$ 基本モードの動画スクリプトを作成せよ。

### 4. 知っている便利なコマンド

現在登録されている変数を調べるコマンド

```
gnuplot> show variables
```

線の種類、太さ、ポイントの種類を調べるコマンド

```
gnuplot> test
```